Rules for integrands of the form $(d\,\text{Trig}[e+f\,x])^m\,(a+b\,(c\,\text{Tan}[e+f\,x])^n)^p$

0: $\int u\,\left(a+b\,\text{Tan}[e+f\,x]^2\right)^p\,dx$ when $a==b$

- **Derivation: Algebraic simplification**
- **Basis:** $1+\text{Tan}[z]^2==\text{Sec}[z]^2$
- **Rule: If** $a==b$, **then**

$$\int u\,\left(a+b\,\text{Tan}[e+f\,x]^2\right)^p\,dx\;\longrightarrow\;\int u\,\left(a\,\text{Sec}[e+f\,x]^2\right)^p\,dx$$

- **Program code:**

```
Int[u_.*(a_+b_.*tan[e_.+f_.*x_]^2)^p_,x_Symbol] :=
  Int[ActivateTrig[u*(a*sec[e+f*x]^2)^p],x] /;
FreeQ[{a,b,e,f,p},x] && EqQ[a,b]
```

1. $\int (d\,\text{Trig}[e+f\,x])^m\,(b\,(c\,\text{Tan}[e+f\,x])^n)^p\,dx$ when $p\notin\mathbb{Z}$

1: $\int u\,(b\,\text{Tan}[e+f\,x]^n)^p\,dx$ when $p\notin\mathbb{Z}\,\wedge\,n\in\mathbb{Z}$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_x\frac{\left(b\,\text{Tan}[e+f\,x]^n\right)^p}{\text{Tan}[e+f\,x]^{n\,p}}==0$
- **Rule: If** $p\notin\mathbb{Z}\,\wedge\,n\in\mathbb{Z}$, **then**

$$\int u\,(b\,\text{Tan}[e+f\,x]^n)^p\,dx\;\longrightarrow\;\frac{b^{\text{IntPart}[p]}\,(b\,\text{Tan}[e+f\,x]^n)^{\text{FracPart}[p]}}{\text{Tan}[e+f\,x]^{n\,\text{FracPart}[p]}}\int u\,\text{Tan}[e+f\,x]^{n\,p}\,dx$$

- **Program code:**

```
Int[u_.*(b_.*tan[e_.+f_.*x_]^n_)^p_,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  (b*ff^n)^IntPart[p]*(b*Tan[e+f*x]^n)^FracPart[p]/(Tan[e+f*x]/ff)^(n*FracPart[p])*
    Int[ActivateTrig[u]*(Tan[e+f*x]/ff)^(n*p),x]] /;
FreeQ[{b,e,f,n,p},x] && Not[IntegerQ[p]] && IntegerQ[n] &&
  (EqQ[u,1] || MatchQ[u,(d_.*trig_[e+f*x])^m_. /; FreeQ[{d,m},x] && MemberQ[{sin,cos,tan,cot,sec,csc},trig]])
```

2: $\int u\,(b\,(c\,\mathrm{Tan}[e+f\,x])^n)^p\,dx$ when $p \notin \mathbb{Z} \land n \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{(b\,(c\,\mathrm{Tan}[e+f\,x])^n)^p}{(c\,\mathrm{Tan}[e+f\,x])^{n\,p}} == 0$

Rule: If $p \notin \mathbb{Z} \land n \notin \mathbb{Z}$**, then**

$$\int (b\,(c\,\mathrm{Tan}[e+f\,x])^n)^p\,dx \longrightarrow \frac{b^{\mathrm{IntPart}[p]}\,(b\,(c\,\mathrm{Tan}[e+f\,x])^n)^{\mathrm{FracPart}[p]}}{(c\,\mathrm{Tan}[e+f\,x])^{n\,\mathrm{FracPart}[p]}}\int (c\,\mathrm{Tan}[e+f\,x])^{n\,p}\,dx$$

- **Program code:**

```
Int[u_.*(b_.*(c_.*tan[e_.+f_.*x_])^n_)^p_,x_Symbol] :=
  b^IntPart[p]*(b*(c*Tan[e+f*x])^n)^FracPart[p]/(c*Tan[e+f*x])^(n*FracPart[p])*
    Int[ActivateTrig[u]*(c*Tan[e+f*x])^(n*p),x] /;
FreeQ[{b,c,e,f,n,p},x] && Not[IntegerQ[p]] && Not[IntegerQ[n]] &&
  (EqQ[u,1] || MatchQ[u,(d_.*trig_[e+f*x])^m_. /; FreeQ[{d,m},x] && MemberQ[{sin,cos,tan,cot,sec,csc},trig]])
```

2. $\int (a+b\,(c\,\mathrm{Tan}[e+f\,x])^n)^p\,dx$

1: $\int \frac{1}{a+b\,\mathrm{Tan}[e+f\,x]^2}\,dx$ when $a \neq b$

Derivation: Algebraic expansion

- **Basis:** $\frac{1}{a+b\,\mathrm{Tan}[z]^2} == \frac{1}{a-b} - \frac{b\,\mathrm{Sec}[z]^2}{(a-b)\,(a+b\,\mathrm{Tan}[z]^2)}$

Rule: If $a \neq b$**, then**

$$\int \frac{1}{a+b\,\mathrm{Tan}[e+f\,x]^2}\,dx \longrightarrow \frac{x}{a-b} - \frac{b}{a-b}\int \frac{\mathrm{Sec}[e+f\,x]^2}{a+b\,\mathrm{Tan}[e+f\,x]^2}\,dx$$

- **Program code:**

```
Int[1/(a_+b_.*tan[e_.+f_.*x_]^2),x_Symbol] :=
  x/(a-b) - b/(a-b)*Int[Sec[e+f*x]^2/(a+b*Tan[e+f*x]^2),x] /;
FreeQ[{a,b,e,f},x] && NeQ[a,b]
```

2: $\int (a+b\,(c\,\text{Tan}[e+f\,x])^n)^p\,dx$ when $(n \mid p) \in \mathbb{Z}\ \lor\ p \in \mathbb{Z}^+ \lor\ n^2 == 4\ \lor\ n^2 == 16$

Derivation: Integration by substitution

Basis: $F[c\,\text{Tan}[e+f\,x]] == \frac{c}{f}\,\text{Subst}\left[\frac{F[x]}{c^2+x^2},\ x,\ c\,\text{Tan}[e+f\,x]\right]\partial_x(c\,\text{Tan}[e+f\,x])$

Note: If $(n \mid p) \in \mathbb{Z}\ \lor\ p \in \mathbb{Z}^+ \lor\ n^2 == 4\ \lor\ n^2 == 16$, **then** $\frac{(a+b\,x^n)^p}{c^2+x^2}$ **is integrable.**

Rule: If $(n \mid p) \in \mathbb{Z}\ \lor\ p \in \mathbb{Z}^+ \lor\ n^2 == 4\ \lor\ n^2 == 16$, **then**

$$\int (a+b\,(c\,\text{Tan}[e+f\,x])^n)^p\,dx \longrightarrow \frac{c}{f}\,\text{Subst}\left[\int \frac{(a+b\,x^n)^p}{c^2+x^2}\,dx,\ x,\ c\,\text{Tan}[e+f\,x]\right]$$

Program code:

```
Int[(a_+b_.*(c_.*tan[e_.+f_.*x_])^n_)^p_,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  c*ff/f*Subst[Int[(a+b*(ff*x)^n)^p/(c^2+ff^2*x^2),x],x,c*Tan[e+f*x]/ff]] /;
FreeQ[{a,b,c,e,f,n,p},x] && (IntegersQ[n,p] || IGtQ[p,0] || EqQ[n^2,4] || EqQ[n^2,16])
```

X: $\int (a+b\,(c\,\text{Tan}[e+f\,x])^n)^p\,dx$

Rule:

$$\int (a+b\,(c\,\text{Tan}[e+f\,x])^n)^p\,dx \longrightarrow \int (a+b\,(c\,\text{Tan}[e+f\,x])^n)^p\,dx$$

Program code:

```
Int[(a_+b_.*(c_.*tan[e_.+f_.*x_])^n_)^p_.,x_Symbol] :=
  Unintegrable[(a+b*(c*Tan[e+f*x])^n)^p,x] /;
FreeQ[{a,b,c,e,f,n,p},x]
```

3. $\int (d \operatorname{Sin}[e+f x])^m \left(a+b (c \operatorname{Tan}[e+f x])^n\right)^p dx$

1: $\int \operatorname{Sin}[e+f x]^m \left(a+b (c \operatorname{Tan}[e+f x])^n\right)^p dx$ when $\frac{m}{2} \in \mathbb{Z}$

Derivation: Integration by substitution

- **Basis:** $\operatorname{Sin}[z]^2 == \frac{\operatorname{Tan}[z]^2}{1+\operatorname{Tan}[z]^2}$
- **Basis: If** $\frac{m}{2} \in \mathbb{Z}$**, then** $\operatorname{Sin}[e+f x]^m F[c \operatorname{Tan}[e+f x]] == \frac{c}{f} \operatorname{Subst}\left[\frac{x^m F[x]}{(c^2+x^2)^{\frac{m}{2}+1}}, x, c \operatorname{Tan}[e+f x]\right] \partial_x (c \operatorname{Tan}[e+f x])$

Rule: If $\frac{m}{2} \in \mathbb{Z}$**, then**

$$\int \operatorname{Sin}[e+f x]^m \left(a+b (c \operatorname{Tan}[e+f x])^n\right)^p dx \longrightarrow \frac{c}{f} \operatorname{Subst}\left[\int \frac{x^m (a+b x^n)^p}{\left(c^2+x^2\right)^{\frac{m}{2}+1}} dx, x, c \operatorname{Tan}[e+f x]\right]$$

Program code:

```
Int[sin[e_.+f_.*x_]^m_*(a_+b_.*(c_.*tan[e_.+f_.*x_])^n_)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  c*ff^(m+1)/f*Subst[Int[x^m*(a+b*(ff*x)^n)^p/(c^2+ff^2*x^2)^(m/2+1),x],x,c*Tan[e+f*x]/ff]] /;
FreeQ[{a,b,c,e,f,n,p},x] && IntegerQ[m/2]
```

2. $\int \text{Sin}[e+f\,x]^m\,(a+b\,\text{Tan}[e+f\,x]^n)^p\,dx$

1: $\int \text{Sin}[e+f\,x]^m\,\left(a+b\,\text{Tan}[e+f\,x]^2\right)^p\,dx$ when $\frac{m-1}{2}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: $\text{Tan}[z]^2 == -1+\text{Sec}[z]^2$

Basis: If $\frac{m-1}{2}\in\mathbb{Z}$**, then** $\text{Sin}[e+f\,x]^m\,F\left[\text{Tan}[e+f\,x]^2\right] == \frac{1}{f}\,\text{Subst}\left[\frac{(-1+x^2)^{\frac{m-1}{2}}\,F[-1+x^2]}{x^{m+1}},\,x,\,\text{Sec}[e+f\,x]\right]\,\partial_x\text{Sec}[e+f\,x]$

Rule: If $\frac{m-1}{2}\in\mathbb{Z}$**, then**

$$\int \text{Sin}[e+f\,x]^m\,\left(a+b\,\text{Tan}[e+f\,x]^2\right)^p\,dx\,\longrightarrow\,\frac{1}{f}\,\text{Subst}\left[\int\frac{\left(-1+x^2\right)^{\frac{m-1}{2}}\,\left(a-b+b\,x^2\right)^p}{x^{m+1}}\,dx,\,x,\,\text{Sec}[e+f\,x]\right]$$

Program code:

```
Int[sin[e_.+f_.*x_]^m_.*(a_+b_.*tan[e_.+f_.*x_]^2)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Sec[e+f*x],x]},
  1/(f*ff^m)*Subst[Int[(-1+ff^2*x^2)^((m-1)/2)*(a-b+b*ff^2*x^2)^p/x^(m+1),x],x,Sec[e+f*x]/ff]] /;
FreeQ[{a,b,e,f,p},x] && IntegerQ[(m-1)/2]
```

2: $\int \text{Sin}[e+f\,x]^m\,(a+b\,\text{Tan}[e+f\,x]^n)^p\,dx$ when $\frac{m-1}{2}\in\mathbb{Z}\bigwedge\frac{n}{2}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: $\text{Tan}[z]^2 == -1+\text{Sec}[z]^2$

Basis: If $\frac{m-1}{2}\in\mathbb{Z}$, **then** $\text{Sin}[e+f\,x]^m\,F\left[\text{Tan}[e+f\,x]^2\right] == \frac{1}{f}\,\text{Subst}\left[\frac{(-1+x^2)^{\frac{m-1}{2}}\,F[-1+x^2]}{x^{m+1}},\,x,\,\text{Sec}[e+f\,x]\right]\partial_x\text{Sec}[e+f\,x]$

Rule: If $\frac{m-1}{2}\in\mathbb{Z}\bigwedge\frac{n}{2}\in\mathbb{Z}$, **then**

$$\int \text{Sin}[e+f\,x]^m\,(a+b\,\text{Tan}[e+f\,x]^n)^p\,dx\;\longrightarrow\;\frac{1}{f}\,\text{Subst}\left[\int\frac{\left(-1+x^2\right)^{\frac{m-1}{2}}\left(a+b\left(-1+x^2\right)^{n/2}\right)^p}{x^{m+1}}\,dx,\,x,\,\text{Sec}[e+f\,x]\right]$$

Program code:

```
Int[sin[e_.+f_.*x_]^m_.*(a_+b_.*tan[e_.+f_.*x_]^n_)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Sec[e+f*x],x]},
  1/(f*ff^m)*Subst[Int[(-1+ff^2*x^2)^((m-1)/2)*(a+b*(-1+ff^2*x^2)^(n/2))^p/x^(m+1),x],x,Sec[e+f*x]/ff]] /;
FreeQ[{a,b,e,f,p},x] && IntegerQ[(m-1)/2] && IntegerQ[n/2]
```

3: $\int (d\,\text{Sin}[e+f\,x])^m\,(a+b\,(c\,\text{Tan}[e+f\,x])^n)^p\,dx$ when $p\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p\in\mathbb{Z}^+$, **then**

$$\int (d\,\text{Sin}[e+f\,x])^m\,(a+b\,(c\,\text{Tan}[e+f\,x])^n)^p\,dx\;\longrightarrow\;\int \text{ExpandTrig}\left[(d\,\text{Sin}[e+f\,x])^m\,(a+b\,(c\,\text{Tan}[e+f\,x])^n)^p,\,x\right]dx$$

Program code:

```
Int[(d_.*sin[e_.+f_.*x_])^m_.*(a_+b_.*(c_.*tan[e_.+f_.*x_])^n_)^p_.,x_Symbol] :=
  Int[ExpandTrig[(d*sin[e+f*x])^m*(a+b*(c*tan[e+f*x])^n)^p,x],x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && IGtQ[p,0]
```

4: $\int (d \, \text{Sin}[e+f\,x])^m \left(a+b\,\text{Tan}[e+f\,x]^2\right)^p \, dx$ when $m \notin \mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

- **Basis:** $\partial_x \frac{(d\,\text{Sin}[e+f\,x])^m \left(\text{Sec}[e+f\,x]^2\right)^{m/2}}{\text{Tan}[e+f\,x]^m} == 0$
- **Basis:** $F[\text{Tan}[e+f\,x]] == \frac{1}{f}\,\text{Subst}\left[\frac{F[x]}{1+x^2},\ x,\ \text{Tan}[e+f\,x]\right] \partial_x \text{Tan}[e+f\,x]$
- **Rule:** If $m \notin \mathbb{Z}$, then

$$\int (d\,\text{Sin}[e+f\,x])^m \left(a+b\,\text{Tan}[e+f\,x]^2\right)^p dx \longrightarrow \frac{(d\,\text{Sin}[e+f\,x])^m \left(\text{Sec}[e+f\,x]^2\right)^{m/2}}{\text{Tan}[e+f\,x]^m} \int \frac{\text{Tan}[e+f\,x]^m \left(a+b\,\text{Tan}[e+f\,x]^2\right)^p}{\left(1+\text{Tan}[e+f\,x]^2\right)^{m/2}} dx$$

$$\longrightarrow \frac{(d\,\text{Sin}[e+f\,x])^m \left(\text{Sec}[e+f\,x]^2\right)^{m/2}}{f\,\text{Tan}[e+f\,x]^m} \text{Subst}\left[\int \frac{x^m \left(a+b\,x^2\right)^p}{\left(1+x^2\right)^{m/2+1}} dx,\ x,\ \text{Tan}[e+f\,x]\right]$$

- **Program code:**

```
Int[(d_.*sin[e_.+f_.*x_])^m_*(a_+b_.*tan[e_.+f_.*x_]^2)^p_,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff*(d*Sin[e+f*x])^m*(Sec[e+f*x]^2)^(m/2)/(f*Tan[e+f*x]^m)*
    Subst[Int[(ff*x)^m*(a+b*ff^2*x^2)^p/(1+ff^2*x^2)^(m/2+1),x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,d,e,f,m,p},x] && Not[IntegerQ[m]]
```

X: $\int (d\,\text{Sin}[e+f\,x])^m \left(a+b\,(c\,\text{Tan}[e+f\,x])^n\right)^p dx$

Rule:

$$\int (d\,\text{Sin}[e+f\,x])^m \left(a+b\,(c\,\text{Tan}[e+f\,x])^n\right)^p dx \longrightarrow \int (d\,\text{Sin}[e+f\,x])^m \left(a+b\,(c\,\text{Tan}[e+f\,x])^n\right)^p dx$$

- **Program code:**

```
Int[(d_.*sin[e_.+f_.*x_])^m_.*(a_+b_.*(c_.*tan[e_.+f_.*x_])^n_)^p_.,x_Symbol] :=
  Unintegrable[(d*Sin[e+f*x])^m*(a+b*(c*Tan[e+f*x])^n)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x]
```

4: $\int (d\,\text{Cos}[e+f\,x])^m\,(a+b\,(c\,\text{Tan}[e+f\,x])^n)^p\,dx$ when $m \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x\left((d\,\text{Cos}[e+f\,x])^m\left(\frac{\text{Sec}[e+f\,x]}{d}\right)^m\right) = 0$

Rule: If $m \notin \mathbb{Z}$, **then**

$$\int (d\,\text{Cos}[e+f\,x])^m\,(a+b\,(c\,\text{Tan}[e+f\,x])^n)^p\,dx \longrightarrow (d\,\text{Cos}[e+f\,x])^{\text{FracPart}[m]}\left(\frac{\text{Sec}[e+f\,x]}{d}\right)^{\text{FracPart}[m]}\int\left(\frac{\text{Sec}[e+f\,x]}{d}\right)^{-m}(a+b\,(c\,\text{Tan}[e+f\,x])^n)^p\,dx$$

Program code:

```
Int[(d_.*cos[e_.+f_.*x_])^m_*(a_+b_.*(c_.*tan[e_.+f_.*x_])^n_)^p_,x_Symbol] :=
  (d*Cos[e+f*x])^FracPart[m]*(Sec[e+f*x]/d)^FracPart[m]*Int[(Sec[e+f*x]/d)^(-m)*(a+b*(c*Tan[e+f*x])^n)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && Not[IntegerQ[m]]
```

5. $\int (d\,\text{Tan}[e+f\,x])^m\,(a+b\,(c\,\text{Tan}[e+f\,x])^n)^p\,dx$

1: $\int (d\,\text{Tan}[e+f\,x])^m\,(a+b\,(c\,\text{Tan}[e+f\,x])^n)^p\,dx$ when $p \in \mathbb{Z}^+ \lor n == 2 \lor n == 4 \lor a == 0$

Derivation: Integration by substitution

Basis: $F[c\,\text{Tan}[e+f\,x]] == \frac{c}{f}\,\text{Subst}\left[\frac{F[x]}{c^2+x^2},\,x,\,c\,\text{Tan}[e+f\,x]\right]\partial_x(c\,\text{Tan}[e+f\,x])$

Rule: If $p \in \mathbb{Z}^+ \lor n == 2 \lor n == 4 \lor a == 0$, **then**

$$\int (d\,\text{Tan}[e+f\,x])^m\,(a+b\,(c\,\text{Tan}[e+f\,x])^n)^p\,dx \longrightarrow \frac{c}{f}\,\text{Subst}\left[\int\left(\frac{d\,x}{c}\right)^m\frac{(a+b\,x^n)^p}{c^2+x^2}\,dx,\,x,\,c\,\text{Tan}[e+f\,x]\right]$$

Program code:

```
Int[(d_.*tan[e_.+f_.*x_])^m_.*(a_+b_.*(c_.*tan[e_.+f_.*x_])^n_)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  c*ff/f*Subst[Int[(d*ff*x/c)^m*(a+b*(ff*x)^n)^p/(c^2+ff^2*x^2),x],x,c*Tan[e+f*x]/ff]] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && (IGtQ[p,0] || EqQ[n,2] || EqQ[n,4] || IntegerQ[p] && RationalQ[n])
```

2: $\int (d\,\text{Tan}[e+f\,x])^m\,(a+b\,(c\,\text{Tan}[e+f\,x])^n)^p\,dx$ when $p\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p\in\mathbb{Z}^+$, then

$$\int (d\,\text{Tan}[e+f\,x])^m\,(a+b\,(c\,\text{Tan}[e+f\,x])^n)^p\,dx \longrightarrow \int \text{ExpandTrig}[(d\,\text{Tan}[e+f\,x])^m\,(a+b\,(c\,\text{Tan}[e+f\,x])^n)^p,\,x]\,dx$$

Program code:

```
Int[(d_.*tan[e_.+f_.*x_])^m_.*(a_+b_.*(c_.*tan[e_.+f_.*x_])^n_)^p_.,x_Symbol] :=
  Int[ExpandTrig[(d*tan[e+f*x])^m*(a+b*(c*tan[e+f*x])^n)^p,x],x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && IGtQ[p,0]
```

X: $\int (d\,\text{Tan}[e+f\,x])^m\,(a+b\,(c\,\text{Tan}[e+f\,x])^n)^p\,dx$

Rule:

$$\int (d\,\text{Tan}[e+f\,x])^m\,(a+b\,(c\,\text{Tan}[e+f\,x])^n)^p\,dx \longrightarrow \int (d\,\text{Tan}[e+f\,x])^m\,(a+b\,(c\,\text{Tan}[e+f\,x])^n)^p\,dx$$

Program code:

```
Int[(d_.*tan[e_.+f_.*x_])^m_.*(a_+b_.*(c_.*tan[e_.+f_.*x_])^n_)^p_.,x_Symbol] :=
  Unintegrable[(d*Tan[e+f*x])^m*(a+b*(c*Tan[e+f*x])^n)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x]
```

6. $\int (d \operatorname{Cot}[e+f x])^m (a+b (c \operatorname{Tan}[e+f x])^n)^p \, dx$ when $m \notin \mathbb{Z}$

1: $\int (d \operatorname{Cot}[e+f x])^m (a+b \operatorname{Tan}[e+f x]^n)^p \, dx$ when $m \notin \mathbb{Z} \land (n \mid p) \in \mathbb{Z}$

Derivation: Algebraic normalization

Basis: If $(n \mid p) \in \mathbb{Z}$**, then** $(a+b \operatorname{Tan}[e+f x]^n)^p == d^{n p} (d \operatorname{Cot}[e+f x])^{-n p} (b+a \operatorname{Cot}[e+f x]^n)^p$

Rule: If $m \notin \mathbb{Z} \land (n \mid p) \in \mathbb{Z}$**, then**

$$\int (d \operatorname{Cot}[e+f x])^m (a+b \operatorname{Tan}[e+f x]^n)^p \, dx \longrightarrow d^{n p} \int (d \operatorname{Cot}[e+f x])^{m-n p} (b+a \operatorname{Cot}[e+f x]^n)^p \, dx$$

Program code:

```
Int[(d_.*cot[e_.+f_.*x_])^m_*(a_+b_.*tan[e_.+f_.*x_]^n_.)^p_.,x_Symbol] :=
  d^(n*p)*Int[(d*Cot[e+f*x])^(m-n*p)*(b+a*Cot[e+f*x]^n)^p,x] /;
FreeQ[{a,b,d,e,f,m,n,p},x] && Not[IntegerQ[m]] && IntegersQ[n,p]
```

2: $\int (d \operatorname{Cot}[e+f x])^m (a+b (c \operatorname{Tan}[e+f x])^n)^p \, dx$ when $m \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \left((d \operatorname{Cot}[e+f x])^m \left(\frac{\operatorname{Tan}[e+f x]}{d}\right)^m\right) == 0$

Rule: If $m \notin \mathbb{Z}$**, then**

$$\int (d \operatorname{Cot}[e+f x])^m (a+b (c \operatorname{Tan}[e+f x])^n)^p \, dx \longrightarrow (d \operatorname{Cot}[e+f x])^{\operatorname{FracPart}[m]} \left(\frac{\operatorname{Tan}[e+f x]}{d}\right)^{\operatorname{FracPart}[m]} \int \left(\frac{\operatorname{Tan}[e+f x]}{d}\right)^{-m} (a+b (c \operatorname{Tan}[e+f x])^n)^p \, dx$$

Program code:

```
Int[(d_.*cot[e_.+f_.*x_])^m_*(a_+b_.*(c_.*tan[e_.+f_.*x_])^n_)^p_,x_Symbol] :=
  (d*Cot[e+f*x])^FracPart[m]*(Tan[e+f*x]/d)^FracPart[m]*Int[(Tan[e+f*x]/d)^(-m)*(a+b*(c*Tan[e+f*x])^n)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && Not[IntegerQ[m]]
```

7. $\int (d\,\text{Sec}[e+f\,x])^m\,(a+b\,(c\,\text{Tan}[e+f\,x])^n)^p\,dx$

1: $\int \text{Sec}[e+f\,x]^m\,(a+b\,(c\,\text{Tan}[e+f\,x])^n)^p\,dx$ when $\frac{m}{2}\in\mathbb{Z}\bigwedge\left((n\mid p)\in\mathbb{Z}\bigvee\frac{m}{2}\in\mathbb{Z}^+\bigvee p\in\mathbb{Z}^+\bigvee n^2==4\bigvee n^2==16\right)$

Derivation: Integration by substitution

- **Basis: If** $\frac{m}{2}\in\mathbb{Z}$**, then** $\text{Sec}[e+f\,x]^m\,F[c\,\text{Tan}[e+f\,x]]==\frac{1}{c^{m-1}\,f}\,\text{Subst}\left[\left(c^2+x^2\right)^{\frac{m}{2}-1}F[x],\,x,\,c\,\text{Tan}[e+f\,x]\right]\partial_x\,(c\,\text{Tan}[e+f\,x])$
- **Note: If** $(n\mid p)\in\mathbb{Z}\bigvee\frac{m}{2}\in\mathbb{Z}^+\bigvee p\in\mathbb{Z}^+\bigvee n^2==4\bigvee n^2==16$**, then** $\left(c^2+x^2\right)^{\frac{m}{2}-1}(a+b\,x^n)^p$ **is integrable.**
- **Rule: If** $\frac{m}{2}\in\mathbb{Z}\bigwedge\left((n\mid p)\in\mathbb{Z}\bigvee\frac{m}{2}\in\mathbb{Z}^+\bigvee p\in\mathbb{Z}^+\bigvee n^2==4\bigvee n^2==16\right)$**, then**

$$\int \text{Sec}[e+f\,x]^m\,(a+b\,(c\,\text{Tan}[e+f\,x])^n)^p\,dx\;\rightarrow\;\frac{1}{c^{m-1}\,f}\,\text{Subst}\left[\int\left(c^2+x^2\right)^{\frac{m}{2}-1}(a+b\,x^n)^p\,dx,\,x,\,c\,\text{Tan}[e+f\,x]\right]$$

Program code:

```
Int[sec[e_.+f_.*x_]^m_*(a_+b_.*(c_.*tan[e_.+f_.*x_])^n_)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff/(c^(m-1)*f)*Subst[Int[(c^2+ff^2*x^2)^(m/2-1)*(a+b*(ff*x)^n)^p,x],x,c*Tan[e+f*x]/ff]] /;
FreeQ[{a,b,c,e,f,n,p},x] && IntegerQ[m/2] && (IntegersQ[n,p] || IGtQ[m,0] || IGtQ[p,0] || EqQ[n^2,4] || EqQ[n^2,16])
```

2. $\int \text{Sec}[e+f x]^m \left(a+b\,\text{Tan}[e+f x]^n\right)^p\,dx$ when $\frac{m-1}{2}\in\mathbb{Z}\bigwedge\frac{n}{2}\in\mathbb{Z}$

1: $\int \text{Sec}[e+f x]^m \left(a+b\,\text{Tan}[e+f x]^n\right)^p\,dx$ when $\frac{m-1}{2}\in\mathbb{Z}\bigwedge\frac{n}{2}\in\mathbb{Z}\bigwedge p\in\mathbb{Z}$

Derivation: Integration by substitution

- **Basis:** $\text{Tan}[z]^2 == \frac{\text{Sin}[z]^2}{1-\text{Sin}[z]^2}$
- **Basis: If** $\frac{m-1}{2}\in\mathbb{Z}$**, then** $\text{Sec}[e+f x]^m F\left[\text{Tan}[e+f x]^2\right] == \frac{1}{f}\,\text{Subst}\left[\frac{F\left[\frac{x^2}{1-x^2}\right]}{(1-x^2)^{\frac{m+1}{2}}},\,x,\,\text{Sin}[e+f x]\right]\partial_x \text{Sin}[e+f x]$
- **Rule: If** $\frac{m-1}{2}\in\mathbb{Z}\bigwedge\frac{n}{2}\in\mathbb{Z}\bigwedge p\in\mathbb{Z}$**, then**

$$\int \text{Sec}[e+f x]^m \left(a+b\,\text{Tan}[e+f x]^n\right)^p dx \longrightarrow \frac{1}{f}\,\text{Subst}\left[\int\frac{\left(b\,x^n+a\left(1-x^2\right)^{n/2}\right)^p}{\left(1-x^2\right)^{\frac{1}{2}(m+n\,p+1)}}\,dx,\,x,\,\text{Sin}[e+f x]\right]$$

Program code:

```
Int[sec[e_.+f_.*x_]^m_.*(a_+b_.*tan[e_.+f_.*x_]^n_)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Sin[e+f*x],x]},
  ff/f*Subst[Int[ExpandToSum[b*(ff*x)^n+a*(1-ff^2*x^2)^(n/2),x]^p/(1-ff^2*x^2)^((m+n*p+1)/2),x],x,Sin[e+f*x]/ff]] /;
FreeQ[{a,b,e,f},x] && IntegerQ[(m-1)/2] && IntegerQ[n/2] && IntegerQ[p]
```

2: $\int \mathrm{Sec}[e+f\,x]^m\,(a+b\,\mathrm{Tan}[e+f\,x]^n)^p\,dx$ when $\frac{m-1}{2}\in\mathbb{Z}\bigwedge\frac{n}{2}\in\mathbb{Z}\bigwedge p\notin\mathbb{Z}$

Derivation: Integration by substitution

- **Basis:** $\mathrm{Tan}[z]^2 == \frac{\mathrm{Sin}[z]^2}{1-\mathrm{Sin}[z]^2}$

- **Basis: If** $\frac{m-1}{2}\in\mathbb{Z}$**, then** $\mathrm{Sec}[e+f\,x]^m\,F\left[\mathrm{Tan}[e+f\,x]^2\right] == \frac{1}{f}\,\mathrm{Subst}\left[\frac{F\left[\frac{x^2}{1-x^2}\right]}{(1-x^2)^{\frac{m+1}{2}}},\,x,\,\mathrm{Sin}[e+f\,x]\right]\partial_x\mathrm{Sin}[e+f\,x]$

- **Rule: If** $\frac{m-1}{2}\in\mathbb{Z}\bigwedge\frac{n}{2}\in\mathbb{Z}\bigwedge p\notin\mathbb{Z}$**, then**

$$\int \mathrm{Sec}[e+f\,x]^m\,(a+b\,\mathrm{Tan}[e+f\,x]^n)^p\,dx \longrightarrow \frac{1}{f}\,\mathrm{Subst}\left[\int\frac{1}{\left(1-x^2\right)^{\frac{m+1}{2}}}\left(\frac{b\,x^n+a\,\left(1-x^2\right)^{n/2}}{\left(1-x^2\right)^{\frac{n}{2}}}\right)^p dx,\,x,\,\mathrm{Sin}[e+f\,x]\right]$$

Program code:

```
Int[sec[e_.+f_.*x_]^m_.*(a_+b_.*tan[e_.+f_.*x_]^n_)^p_.,x_Symbol] :=
  With[{ff=FreeFactors[Sin[e+f*x],x]},
  ff/f*Subst[Int[1/(1-ff^2*x^2)^((m+1)/2)*((b*(ff*x)^n+a*(1-ff^2*x^2)^(n/2))/(1-ff^2*x^2)^(n/2))^p,x],x,Sin[e+f*x]/ff]] /;
FreeQ[{a,b,e,f,p},x] && IntegerQ[(m-1)/2] && IntegerQ[n/2] && Not[IntegerQ[p]]
```

3: $\int (d\,\mathrm{Sec}[e+f\,x])^m\,(a+b\,(c\,\mathrm{Tan}[e+f\,x])^n)^p\,dx$ when $p\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p\in\mathbb{Z}^+$**, then**

$$\int (d\,\mathrm{Sec}[e+f\,x])^m\,(a+b\,(c\,\mathrm{Tan}[e+f\,x])^n)^p\,dx \longrightarrow \int \mathrm{ExpandTrig}[(d\,\mathrm{Sec}[e+f\,x])^m\,(a+b\,(c\,\mathrm{Tan}[e+f\,x])^n)^p,\,x]\,dx$$

Program code:

```
Int[(d_.*sec[e_.+f_.*x_])^m_.*(a_+b_.*(c_.*tan[e_.+f_.*x_])^n_)^p_.,x_Symbol] :=
  Int[ExpandTrig[(d*sec[e+f*x])^m*(a+b*(c*tan[e+f*x])^n)^p,x],x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && IGtQ[p,0]
```

4: $\int (d \operatorname{Sec}[e+f x])^m \left(a+b \operatorname{Tan}[e+f x]^2\right)^p \, dx$ when $m \notin \mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

- **Basis:** $\partial_x \frac{(d \operatorname{Sec}[e+f x])^m}{\left(\operatorname{Sec}[e+f x]^2\right)^{m/2}} == 0$
- **Basis:** $F[\operatorname{Tan}[e+f x]] == \frac{1}{f} \operatorname{Subst}\left[\frac{F[x]}{1+x^2}, x, \operatorname{Tan}[e+f x]\right] \partial_x \operatorname{Tan}[e+f x]$
- **Rule: If** $m \notin \mathbb{Z}$**, then**

$$\int (d \operatorname{Sec}[e+f x])^m \left(a+b \operatorname{Tan}[e+f x]^2\right)^p \, dx \to \frac{(d \operatorname{Sec}[e+f x])^m}{\left(\operatorname{Sec}[e+f x]^2\right)^{m/2}} \int \left(1+\operatorname{Tan}[e+f x]^2\right)^{m/2} \left(a+b \operatorname{Tan}[e+f x]^2\right)^p \, dx$$

$$\to \frac{(d \operatorname{Sec}[e+f x])^m}{f \left(\operatorname{Sec}[e+f x]^2\right)^{m/2}} \operatorname{Subst}\left[\int \left(1+x^2\right)^{m/2-1} \left(a+b x^2\right)^p \, dx, x, \operatorname{Tan}[e+f x]\right]$$

- **Program code:**

```
Int[(d_.*sec[e_.+f_.*x_])^m_*(a_+b_.*tan[e_.+f_.*x_]^2)^p_,x_Symbol] :=
  With[{ff=FreeFactors[Tan[e+f*x],x]},
  ff*(d*Sec[e+f*x])^m/(f*(Sec[e+f*x]^2)^(m/2))*
    Subst[Int[(1+ff^2*x^2)^(m/2-1)*(a+b*ff^2*x^2)^p,x],x,Tan[e+f*x]/ff]] /;
FreeQ[{a,b,d,e,f,m,p},x] && Not[IntegerQ[m]]
```

X: $\int (d \operatorname{Sec}[e+f x])^m (a+b (c \operatorname{Tan}[e+f x])^n)^p \, dx$

- **Rule:**

$$\int (d \operatorname{Sec}[e+f x])^m (a+b (c \operatorname{Tan}[e+f x])^n)^p \, dx \to \int (d \operatorname{Sec}[e+f x])^m (a+b (c \operatorname{Tan}[e+f x])^n)^p \, dx$$

- **Program code:**

```
Int[(d_.*sec[e_.+f_.*x_])^m_.*(a_+b_.*(c_.*tan[e_.+f_.*x_])^n_)^p_.,x_Symbol] :=
  Unintegrable[(d*Sec[e+f*x])^m*(a+b*(c*Tan[e+f*x])^n)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x]
```

8: $\int (d \operatorname{Csc}[e+f x])^m (a+b (c \operatorname{Tan}[e+f x])^n)^p \, dx$ when $m \notin \mathbb{Z}$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_x \left((d \operatorname{Csc}[e+f x])^m \left(\frac{\operatorname{Sin}[e+f x]}{d}\right)^m\right) == 0$
- **Rule: If** $m \notin \mathbb{Z}$**, then**

$$\int (d \operatorname{Csc}[e+f x])^m (a+b (c \operatorname{Tan}[e+f x])^n)^p \, dx \longrightarrow (d \operatorname{Csc}[e+f x])^{\operatorname{FracPart}[m]} \left(\frac{\operatorname{Sin}[e+f x]}{d}\right)^{\operatorname{FracPart}[m]} \int \left(\frac{\operatorname{Sin}[e+f x]}{d}\right)^{-m} (a+b (c \operatorname{Tan}[e+f x])^n)^p \, dx$$

- **Program code:**

```
Int[(d_.*csc[e_.+f_.*x_])^m_*(a_+b_.*(c_.*tan[e_.+f_.*x_])^n_)^p_,x_Symbol] :=
  (d*Csc[e+f*x])^FracPart[m]*(Sin[e+f*x]/d)^FracPart[m]*Int[(Sin[e+f*x]/d)^(-m)*(a+b*(c*Tan[e+f*x])^n)^p,x] /;
FreeQ[{a,b,c,d,e,f,m,n,p},x] && Not[IntegerQ[m]]
```